
Работа с файлами в ИСР Turbo Pascal 7.0 / СОШ №1134 г. Москвы / Друца А.В. / 2005 г.

I. Файлы, общие понятия.

Файл представляет собой поименованную последовательность однотипных элементов, размещённых на внешнем устройстве, чаще всего, на диске. Далее в этом пособии подразумевается, что файл находится именно на диске.

Для выполнения операций с конкретным файлом, размещённым на диске, в программе обычно используется так называемая файловая переменная (логический файл). Файловая переменная после её описания связывается с некоторым файлом, в результате чего операции, выполняемые над ней, приводят к соответствующим изменениям в этом файле. После завершения всех операций связь между файловой переменной и файлом разрывается. Далее файловую переменную можно связать с другим файлом этого же типа.

В зависимости от типа элементов различают текстовые, типизированные и нетипизированные файлы.
· Текстовый файл содержит строки символов переменной длины. О таких файлах будет рассказано в разделе IV настоящего пособия.
· Типизированный файл составляют элементы указанного типа (кроме файлового).
· В нетипизированном файле находятся элементы, тип которых не указан. В таком случае работа с данными в файле осуществляется на физическом уровне. В нашем курсе работу с такими файлами мы рассматривать не будем.

Тем не менее, независимо от типа каждый файл – это некоторый набор чисел (байтов) на внешнем устройстве.
II. Работа с типизированными файлами.

Итак, в общем случае файловые переменные, предназначенные для работы с типизированными файлами, могут объявляться следующим образом:
VAR
   Имя_файловой_переменной : file of Тип_данных_элемента;


При этом имя переменной типа file выбирается произвольным образом, подобно тому, как выбирается любой другой идентификатор. Тип данных элемента типа file может быть любым из допустимых в TP 7.0, за исключением типа файлов. Примеры:
VAR

   F1 : file of integer; {для работы с файлами, содержащими целые числа}
   F2 : file of real; {для работы с файлами, содержащими действительные числа}
   InputFile : file of char; {для работы с файлами, содержащими символы}

Для связывания файловой переменной с конкретным файлом, хранящимся на диске, в TP 7.0 предназначена процедура assign(F,FileName), где F – имя файловой переменной, а FileName – имя строковой переменной, в которой хранится имя файла на диске. В качестве параметра имени файла в этой процедуре может фигурировать и текстовая константа. Примеры использования:
   filename:=’data.aaa’;

   assign(F1, filename); {связываем с файлом, имя которого хранится в filename}
   assign(F2, ’myfile.z’); {связываем с файлом, находящимся в каталоге программы}
   assign(InputFile, ’D:\Users\files\TopSecret.CIA’); {указано полное имя файла}

Процедура reset(F) открывает существующий файл для последующего чтения из него данных (F – имя файловой переменной, связанной с этим файлом при помощи процедуры assign). Пример:
   assign(F2, ’myfile.z’); {связываем с файлом, находящимся в каталоге программы}
   reset(F2); {открываем файл myfile.z для чтения}

Процедура rewrite(F) создаёт новый файл и открывает его для записи (F – имя файловой переменной, связанной с этим файлом при помощи процедуры assign). Если файл с таким именем уже существует, его содержимое стирается, а сам файл открывается заново. Пример:
   assign(F2, ’myfile.z’); {связываем с файлом, находящимся в каталоге программы}
   rewrite(F2); {открываем файл myfile.z для записи}

Процедура close(F) закрывает открытый ранее файл, связанный с указанной в качестве параметра файловой переменной. Пример:

   close(F2); {закрываем файл, имя которого связано с файловой переменной F2}

Обратите внимание на то, что при попытке закрыть с помощью процедуры close файл, который не был до этого открыт или который уже закрыт, возникает ошибочная ситуация. Если Вы хотите избежать потерь информации или ситуации некорректного ввода/вывода, обязательно закрывайте все ранее открытые файлы в конце программы!

Используя логическую функцию EOF(F) (End Of File – конец файла), в процессе считывания информации Вы можете проверить, достигнут ли конец файла, т.е. находится ли указатель  файла за последним элементом или нет (F – имя файловой переменной, связанной с открытым для чтения файлом). Если в файле прочитаны все элементы, то функция вернёт значение true. Примеры использования данной функции смотрите ниже.

Для чтения данных из файла используется процедура read(F,x), где F – имя файловой переменной, связанной с файлом, открытым для чтения, x – переменная того же типа, что и тип данных элементов, файловой переменной F. В результате выполнения данной процедуры,  в переменную x запишется очередной элемент файл, а указатель, показывающий позицию последнего прочитанного элемента, увеличится на единицу (при открытии файла, этот указатель стоит перед первым элементом файла). При попытке чтения из файла, в котором достигнут его конец, возникает ошибочная ситуация. Поэтому при чтении файла, количество элементов в котором не известно заранее, рекомендуется использовать функцию eof. Примеры:
   read(F1,x); {читаем очередной элемент из файла F1}
   read(F1,x,y); {читаем два последующих элемента из файла F1}

Для записи данных в файл используется процедура write(F,x), где F – имя файловой переменной, связанной с файлом, открытым для записи, x – переменная того же типа, что и тип данных элементов, файловой переменной F. В результате выполнения данной процедуры, значение, хранящееся в переменной x, запишется в конец файла. Примеры:

VAR F1 : file of real;  x, y : real;

   {.......}

   write(F1,x); {записываем значение x в файл F1}
   write(F1,x,y,x+y); { записываем значение x, y и x+y в файл F1}

Подводя итог, укажем ещё раз, какие этапы включает работа с файлами: 

1. связывание файловой переменной с конкретным файлом (процедура assign);
2. открытие файла для чтения или записи (процедуры reset, rewrite, соотв.);
3. чтение (запись) файла при помощи процедуры read (write);

4. закрытие файла – процедура close.
III. Решение задач на работу с типизированными файлами.

Решение задач на обработку файлов имеет много общего с решением задач на одномерные массивы. Тем не менее, при написании программы необходимо учитывать два существенных отличия между файлами и массивами: 
· количество элементов в файле не объявляется;
· в файле не возможно получить доступ к элементу по его порядковому номеру, используя стандартный набор процедур и функций.

Поэтому, при решении задач на обработку всех элементов в созданном ранее файле, в программе следует организовать цикл, который за один шаг будет считывать и обрабатывать очередной элемент файла. Условие окончания работы этого цикла –  достижение конца файла. Приведём несколько примеров решения задач.
Пример 1: Определить сумму отрицательных элементов в файле.

PROGRAM Ex1;

VAR
   F : file of integer;

   x, S : integer;

   FileName : string;

BEGIN
   writeln(’Введите имя файла:’);

   readln(FileName); {Читаем имя файла с клавиатуры}
   assign(F, FileName); {связываем файловую переменную с файлом}
   reset(F); {открываем файл для чтения}
   S:=0;

   while (not EOF(F)) do begin {Цикл работает пока не встретим конец файла}
      read(F, x); {Читаем очередной элемент из файла}
      if (x<0) then S:=S+x;

   end;
   close(F); {Закрываем файл}
   writeln(’Сумма отрицательных элементов файла равна ’, S);
   readln;

END.
Пример 2: Все положительные элементы данного файла сохранить в одном файле, а отрицательные – в другом.
PROGRAM Ex2;

VAR
   F, Fpos, Fneg : file of real;

   x : real;

   FileName : string;

BEGIN

   writeln(’Введите имя файла:’);

   readln(FileName); 

   assign(F, FileName); assign(Fpos, ’pos.out’); assign(Fneg, ’neg.out’);
   reset(F); rewrite(Fpos); rewrite(Fneg);
   while (not EOF(F)) do begin
      read(F, x);
      if (x>0) then write(Fpos, x);

      if (x<0) then write(Fneg, x);
   end;
   close(F); close(Fpos); close(Fneg);

   readln;
END.

Для закрепления материала решите следующие задачи: 5.1 – 5.13 («Задачи по программированию» Яворский Р.Э.).

IV. Работа с текстовыми файлами.

Как уже говорилось ранее, текстовый файл содержит строки символов переменной длины. Поэтому работа с ними аналогична работе с типизированными файлами, но с двумя отличиями. Во-первых, файловая переменная текстового файла объявляется таким образом:

VAR
   Имя_файловой_переменной : text;


А, во-вторых, при чтении и записи строк следует применять процедуры readln(F,st) и writeln(F,st), соответственно.


Следует заметить, что из текстовых файлов (в текстовые файлы) можно считывать (записывать) данные и других типов, таких как integer, real, char и др. Вывод таких данных в файл аналогичен выводу их на экран (выведенные данные, просто, будут попадать не на экран ЭВМ, а в текстовый файл). Например:
   x:=55; y:=3; writeln(TextFile, ’x=’, x, ’ y=’, y, ’ x+y=’, x+y);
После выполнения этих операторов в файле будет находиться строка ’x=55 y=3 x+y=58’.


Чтение простых типов данных из текстового файла продемонстрируем на примере. Пусть в файле a.in находится две строки: ’123 3.14’ и  ’-17 J’.
VAR
   TextFile : text;

   x, y: integer; r : real; d : char;

BEGIN
   assign(TextFile, ’a.in’);

   reset(TextFile);
   read(TextFile, x, r);

   read(TextFile, y);

   read(TextFile, d); {сначала прочитали пробел после ’7’}
   read(TextFile, d); {затем, букву ’J’}
   close(TextFile);

   writeln(x, ’ ’, r, ’ ’, y, ’ ’, d);

   readln;

END.
После выполнения этой программы на экране ЭВМ Вы увидите: ’123  3.14 -17 J’.
Для закрепления материала решите следующие задачи: 5.14 – 5.26 («Задачи по программированию» Яворский Р.Э.).
________________________


Настоящее пособие, указанный выше задачник и другие материалы можно найти на «Неофициальном сайте школы №1134»: http://school-1134.narod.ru (См. раздел «Учёба», далее раздел «Материалы»).

